

Paper 149-28


Creating Drill-Down Graphs Using SAS/Graph® and the Output Delivery System  
 Curtis A. Smith, Defense Contract Audit Agency, La Mirada, CA.

**ABSTRACT**



Using the SAS/Graph procedures and the Output Delivery System (ODS) you can output your graphs to web-enabled HTML files. Using the flexibility of ODS, you can include the graphs as static GIF or JPEG files, or create interactive, customizable graphs for web distribution using ActiveX controls and JAVA applets. This is really cool stuff. With a few little tricks, you can go a step below a simple HTML graph. Yes, you can add drill-down capability to your HTML graphs. Using this capability, you can create graphs that drill-down to new graphs that display more of the details in your data. Hard to believe? In this paper the author will explore the few SAS statement options necessary to make HTML drill-down graphs a reality.

**INTRODUCTION**

With SAS/GRAPH and ODS, you can easily produce graphs for distribution, including on the web, that are very user-friendly. Web enabling your graphs is simple, if you are running SAS version 8. All you have to do is to use the Output Delivery System and set a few options. You can chose from a static image, an interactive JAVA applet format, or an interactive ActiveX control format. The static image format will create an HTML document and an associated GIF or JPEG file. Below is an example of a JAVA enabled graph.



Is there more? You bet. Right-click with your mouse over the graph. Look at the menu choices given to you. There are many options you can set and change to alter the way the graph looks.



There are many really cool features available to you and your users when you create HTML graphs using JAVA applets and ActiveX controls. Look what happens when you place your cursor over the bars of the graph. Information about the bar is displayed. SAS is not running here. This is just your browser!



This is cool stuff!

If you read my other WUSS 2002 paper “Using SAS/Graph and ODS to Web Enabling Your Graphs” then how to create these graphs is old hat. So, I am going to assume you know how to create these. From here we want to go further. Yes, we want to be able to drill down into the details of the graphs we create. Doing so will provide more interactive information to our users. Believe me, if I can do this, so can you.

**IMPORTANT NOTE!**  
 The JAVA and ActiveX graphs require applets and controls to be loaded on the PC to display the graphs. These applets are installed with SAS. However, when distributing the graphs to users who do not have SAS installed on their PC, the applets and controls will need to be installed. The applets and controls contained on a separate CD with the SAS version 8 software and can also be downloaded from the SAS web site.

## SAS ODS CODE

My purpose for this paper is to discuss at length the SAS code necessary to create drill-down web enabled graphs using ODS. To that end, I will concentrate on the ODS code and related options necessary to pull this off. I won't discuss at length all of the SAS/GRAPH code that I use to produce the graphs. All of the key SAS/GRAPH code will be shown, but some of it will not be discussed. (The complete code for the drill-down charts is at the end of the paper.) For my example, I will produce a series of bar charts depicting Major League Baseball home run statistics by team, league, and division for the 1986 season. My main graph will show the home run statistics grouped by league and subgrouped by division. I want the user to be able to press on one of the division bars and drill down to a graph that shows team statistics for that division.

To start, take a look at the ODS code and related options.

```
filename odsout 'c:\wuss 2002';
goptions reset=global pixels=650
          ypixels=450 device=java;
ods html body='baseball_body.html'
         (title="ODS Drill-Down Chart")
         frame='baseball_frame.html'
         contents='baseball_contents.html'
         path=odsout
         gtitle gfootnote style=statdoc
         gpath='c:\wuss 2002'
         parameters=("ddlevel1"
                    "disabledrilldown"="true"
                    "drilldownfunction"=""
                    "drilldownmode"="any"
                    "drillpattern"=""
                    "drilltarget"=""
                    "helplocation"="default"
                    "locale"=""
                    "patternstrip"="false");
ods html close;
```

This code begins with the familiar ODS HTML statement. I have directed the output to a body file called "C:\wuss 2002\baseball\_body.html", and have added a title to the HTML window "ODS Drill-Down Chart". I have also created similar named files for the HTML contents and frame. You might notice that I did not specify the location of the files in the ODS HTML statement. Instead I used the PATH option that refers to the path I defined in a Filename statement. And, I have added the familiar STYLE option and selected my favorite style, "STATDOC". The rest of the stuff on the ODS statement probably looks unfamiliar. We will spend some of our time looking at these options.

There are two approaches we can take to make drill-down graphs. The first method does not use most of the ODS parameter= options shown in the code above. This first method does work with any device we wish to use (more on this in a minute). There are four pieces of SAS code that we need to make this first method work: (1) a GOPTIONS statements with the DEVICE= options, (2) some ODS statement options, (3) the HTML and HTML\_Legend options for the SAS/GRAPH procedure we use, (4) and a little Data Step trick. The second method only works with the JAVA device and only requires the ODS statement options shown above. This second method creates drill-down functionality based upon the values stored in the variables defined in the SAS/GRAPH procedure. I will mention this second method briefly at the end of the paper.

Let's look at the four pieces of SAS code we need to make this first method work. First I will tackle the GOPTIONS statement with the

DEVICE= options, because it is the most straightforward. Then, I address the many ODS statement options. Then, I will address the SAS/GRAPH procedure options and the Data Step trick (as they work hand in hand).

## IMPORTANT GOPTIONS

There are several device choices you can make that will determine the type of output you will get. The essential choices are ACTIVEX, JAVA, HTML, GIF, and JPEG. If you select DEVICE=ACTIVEX, SAS will create an HTML file using ActiveX controls. If you select DEVICE=JAVA, SAS will create an HTML file using JAVA applets. The ActiveX and JAVA enabled graphs will look and behave like the examples we saw at the beginning of the paper. If you select DEVICE=HTML or GIF, SAS will create an HTML file with a link to a GIF file, with the GIF file being a static image of the graph. If you select DEVICE=JPEG, SAS will create an HTML file with a link to a JPEG file, with the JPEG file being a static image of the graph.

While not critical for our drill-down, the GOPTIONS statement with the XPIXEL= and YPIXEL= options defines the number of pixels used for the X and Y axis of the graph. The larger these are, the bigger the graphics image will be. Values of 650 and 450 for XPIXEL and YPIXEL, respectively, seem to work well.

## ODS STATEMENT OPTIONS

There are a few important options within the ODS statement that you may not be familiar with. We will examine these in depth. But, let's start with the easier stuff.

### IMPORTANT NOTE!

I have found documentation on these options difficult to find. According to the SAS Institute, the only place to find documentation is in the SAS for Windows on-line help. Here's how to find the help. While in SAS for Windows, from the menus, select Help  $\circ$  SAS System Help. Then select the SEARCH tab and search on "applet parameters". From the list of found entries, select "GraphApplet Parameters."

## THE GPATH OPTION

The GPATH option is important if you are using DEVICE=HTML, GIF, or JPEG. This option tells SAS/GRAPH where to store the GIF or JPEG file(s) it will create and associate with the HTML file(s) it will also create. You will probably want to specify the same location as the HTML file defined as the ODS output file. I have found that under SAS for Windows if I do not use the GPATH option when creating a GIF or JPEG graph that SAS/GRAPH will place the graphics image in the SAS root folder. I have also found that without the GPATH option and if the HTML and the GIF or JPEG file(s) are not created in the same location, there will be no link in the HTML file to the graphics image. When this happens, your browser will display the icon it uses in place of a graphics file that it cannot find. And when this happens, it won't help to later move the graphics file to the same location as the HTML file, because there is no linkage in the HTML file to the graphics file. If you do specify a GPATH when creating a GIF or JPEG file you can specify a location other than the location of the HTML file. Using the GPATH option causes SAS/GRAPH to include in the HTML file the linkage to the graphics file. But, I have found that if you use the PATH option with a Fileref, then you don't need the GPATH option, as long as you want the GIF or JPEG file in the same location as the Fileref location.

**THE PARAMETERS= OPTION**

This option is the big one - take a deep breath. There are a great many things you can control and customize about your graph and its associated HTML file with the parameters= option. The form of the parameters= option is:

```
parameters=("parameter-name"="parameter-value")
```

The SAS documentation noted earlier states that the parameter-name is not case sensitive, but the parameter-value is case sensitive for some of the parameters. So, if one doesn't seem to behave the way you expect, check for case sensitivity. Most of the parameters that you can use to control the ODS output I have covered in my aforementioned paper. So, I will not repeat the information here. Following are the parameters that can be used to control the drill-down capabilities.

**DdLeveln**

Specifies internal drill-down level specifications.

**DisableDrilldown**

Specifies that the internal JAVA drill-down should be disabled for this graph. The value of the parameter does not matter, but a value must be specified as different browsers handle null values differently. Its presence indicates that drill-down should be disabled. Drill-down can again be enabled by the user using the popup menu. *When using the SAS/GRAPH procedure options HTML and HTML\_Legend (more on these soon) to create drill-down graphs, then the internal JAVA drill-down must be disabled to get predictable results.*

**DrillDownFunction | DrillFunc**

Specifies the name of the JavaScript function to call when DrillDownMode is set to "Script".

**DrillDownMode**

Specifies the drill-down mode to use when drilling down. The default is "Any". Valid values are as follows.

Any	In this mode, the graph first attempts to drill-down using "Local" mode. If that fails, it attempts to drill-down using JavaScript ("Script"). If both these fail, the graph attempts to drill-down using "URL" mode. If all these fail, the graph attempts to drill-down using "HTML" mode.
HTML	This mode uses a substitution pattern to dynamically generate a URL based on the selected chart elements.
Local	This mode constructs a new chart based on existing data within the graph and replaces the old chart with a new chart containing the drill-down information.
Script	This mode invokes the user specified JavaScript function with information about the selected chart elements. The user function is expected to perform the requested drill-down action.
URL	This mode is provided for compatibility with previous SAS/Graph drivers such as the GIF driver. It supports the old HTML= drill-down syntax.

**DrillPattern**

Specifies the drill-down pattern when drill-down mode is HTML.

**DrillTarget**

Specifies the drill-down window target to use when drilling down using HTML mode. The default is "\_blank".

**HelpLocation**

Specifies the URL from which the graph is to display help. The default is the help location for a particular applet release that SAS Institute provides on its external Web site. The parameter can be either "Default" or a valid URL.

**Locale**

Specifies the language and country to use when displaying locale-sensitive strings. The default is to use the region settings of the browser's host machine. This parameter can be used to override the browser's default locale.

**PatternStrip**

Specifies that preceding and trailing white space are to be removed from drill-down substitution patterns before adding to them the generated drill-down string. A value of "True" causes the removal of the white spaces; a value of "False" leaves the white spaces. The default is "False."

*Remember, if you are going to create drill-down graphs using the first method we have been discussing, then internal JAVA drill-down must be disabled to get predictable results.*

**SAS/GRAPH OPTIONS**

There are, of course, many statements and options to tweak your graph output to your liking. There are two very critical ODS related options that have been added to the SAS/GRAPH procedures. These are the HTML and HTML\_Legend options. It is these two options that make the drill-down capability work. These options are defined by SAS as follows:

**HTML=variable**

Identifies the variable in the input data set whose values create links in the HTML file created by the ODS HTML statement. These links are associated with an area of the chart and point to the data or graph you wish to display when the user drills down on the area.

**HTML\_LEGEND=variable**

Identifies the variable in the input data set whose values create links in the HTML file created by the ODS HTML statement. These links are associated with a legend value and point to the data or graph you wish to display when the user drills down on the value.

So, what does this mean? Here's how these work. If you want your user to be able to drill-down by pressing on a portion of the graph (a bar for example) then you set the value of the HTML option to a variable name in the SAS data set that will be used to create the graph. In the SAS data set in that variable, you will already have stored a string value that is the name of the HTML file that you want to call when the user presses the appropriate portion of the graph. If you want your user to drill down when pressing a portion of the legend then you follow the same process but use the HTML\_Legend option. Sound confusing?

```
data baseball;
  set data.baseball;
  length divisiondrill $ 40;
  if league='American' and
     division='West' then
    divisiondrill=
      'HREF="divisionALwest_body.html"';
  else if league='American' and
     division='East' then
    divisiondrill=
      'HREF="divisionALEast_body.html"';
  else if league='National' and
     division='West' then
    divisiondrill=
      'HREF="divisionNLwest_body.html"';
  else if league='National' and
     division='East' then
    divisiondrill=
      'HREF="divisionNLEast_body.html"'; run;
```

Take a look above at the Data Step I used to create the variable in the SAS data set and then the SAS/GRAPH options.

In this Data Step I am creating a work SAS data set named baseball by reading the permanent SAS data set also named baseball. (I generated this sample SAS data set using SAS/ASSIST.) The SAS data set contains baseball statistics for players of the 1986 season. Each row contains, among other things, the team, league, and division to which the player belongs. The point of the Data Step is to create a new file containing the variable named DIVISIONDRILL. That variable will be filled with one of four values, one for each of the combinations of league and division (American League Eastern Division, American League Western Division, National League Eastern Division, National League Western Division). I assigned one of the four values based on the values of the two variables LEAGUE and DIVISION.

Next, take a look at the SAS/GRAPH code.

```
proc gchart data=baseball;
  vbar3d division / discrete
  sumvar=no_home
  group=league
  subgroup=division
  legend=legend1
  html=divisiondrill
  html_legend=divisiondrill;
run;
```

If you are familiar with the GCHART procedure, most of this stuff looks familiar. (Of course, I have left out much of the code that we need to make the chart look good. And, of course, you can use other SAS/GRAPH procedures.) You will find in this example the HTML and HTML\_Legend options. Notice that they refer to the variable DIVISIONDRILL that we created in the work.baseball SAS data set. So, what is this option going to do for us? Well, when the GCHART procedure creates a bar chart for us, it will use the value in the variable DIVISIONDRILL to specify the name of another HTML file we will later create. This other HTML file will be linked and executed when a user presses on the bar or legend in our bar chart.

Now, take a look at the ODS statement and SAS/GRAPH code for one of the four subordinate bar charts that we will create. These four bar charts will be the charts for the combinations of the leagues and divisions.

```
ods html body='divisionALwest_body.html'
  path=odsout;
where league='American' and
  division='West';
  vbar3d team / sumvar=no_home
  subgroup=team
  shape=PRISM patternid=subgroup;
run;
```


Notice carefully that the name of the HTML file that I am creating exactly matches one of the values we placed in the DIVISIONDRILL variable. Also, notice that while we have a separate ODS HTML statement for this separate HTML file, we do not use another GCHART procedure. We simply issue another vbar3d statement to create another chart within the same execution of the GCHART procedure. Also, note the very important WHERE statement. It is here that we create the subordinate graph, in this case, for only the

American league and the Western (West) division. You can create drill-down graphs to any level you like. You just have to create more variables in your input SAS data set to hold the values of the HTML files you want to link to the graph areas.

The SAS Online documentation for SAS/GRAPH has a far more involved example of drill-down graphs.

## RESULTS

To the code we have already seen, you can simply add some trusty LEGEND, AXIS, TITLE, and FOOTNOTE statements, and use a graphing procedure, such as the GCHART procedure. Let's see what we can do to a drill-down bar chart.




Nice chart. Moving the mouse over any one of the four bars changes the mouse shape to a hand (or, whatever Windows scheme you are using). Once your user gets a hand, your user can press the mouse button, and like magic, your user will drill down into another, related graph.

Looking at the source for the HTML file above, you will see the

```
<MAP NAME="idx_map">
  <AREA SHAPE="POLY"
  HREF="divisionNLwest_body.html"
  COORDS="488,506,488,278,508,263,548,263,548,491,528,
  506 " >
  <AREA SHAPE="POLY"
  HREF="divisionNLeast_body.html"
  COORDS="448,506,448,289,468,274,508,274,508,491,488,
  506 " >
  <AREA SHAPE="POLY"
  HREF="divisionALwest_body.html"
  COORDS="368,506,368,160,388,145,428,145,428,491,408,
  506 " >
  <AREA SHAPE="POLY"
  HREF="divisionALLeast_body.html"
  COORDS="328,506,328,151,348,136,388,136,388,491,368,
  506 " ></MAP>
<IMG SRC="c:\wuss 2002\mlb1.gif" border="0"
USEMAP="#idx_map">
```

HTML tags in the code above.

I know it's a bit cluttered, but notice the same four strings we specified in the DIVISIONDRILL variable. SAS placed the values of that variable in the HTML file to trigger a hyperlink to the four other HTML graph files we created. Below is one of the four division



charts.

Beautiful, isn't it? SAS/GRAPH and ODS work together so well.

But, what about that second method I mentioned that only requires the use of all those ODS parameter= values. Well, you can exclude the disabledrilldown option, use DEVICE=JAVA, don't create a SAS data set with the HTML strings, and don't use the SAS/GRAPH HTML and HTML\_Legend options. Then, SAS will use the ODS parameter= options to create internal drill-down functionality. The drill-down capability will be within the same graph. The SAS documentation on the parameter= options is meager. But I did find some additional help on the SAS web site in a file called "javaapplets.html". Here's an important quote from that help document.

The drill-down modes in the SAS/GRAPH Applets for Java are data-driven, which means that the target addresses that they go to are constructed dynamically, based on the data values contained in a selected graph element. This contrasts with the drill-down capabilities of GIF images [the method I just presented], which are based on pixel areas that are linked to target locations that are statically defined in an HTML file.

However, I have not been able to get predictable results using this method. Maybe a future paper...



#### REALLY IMPORTANT NOTE!

There are many SAS/GRAPH Options and other statement options that are just not supported by the JAVA and ActiveX devices. Using the unsupported options will not generate an error in the SAS log, but will generate a note in the log. There are too many unsupported options to document in this paper. But, the SAS Institute does have technical support documents TS601 and TS602 that detail the unsupported options. You can find these documents at:

<http://ftp.sas.com/techsup/download/technote/ts601.html>  
<http://ftp.sas.com/techsup/download/technote/ts602.html>

#### CONCLUSION

As I mentioned in the beginning of the paper- if I can create drill-down graphs with SAS/GRAPH and ODS, then anyone can. This is actually pretty simple stuff. Give it a try. You are just a few minutes away from creating your own examples of web enabled drill-down graphs. My SAS code is shown on the next page. Due to space constraints, I only show the code for one of the four subordinate bar charts. The other three follow the same pattern.

#### REFERENCES

SAS JAVA applets and ActiveX controls available at:

[http://www.sas.com/apps/apps/demosdownloads/activex\\_prod\\_81\\_downarea.jsp?productID=100037](http://www.sas.com/apps/apps/demosdownloads/activex_prod_81_downarea.jsp?productID=100037)

A handy SAS Institute document describing web enabling graphs, especially using drill-down capabilities can be found at:

[http://www.sas.com/apps/apps/demosdownloads/activex\\_prod\\_81\\_downarea.jsp?productID=100038](http://www.sas.com/apps/apps/demosdownloads/activex_prod_81_downarea.jsp?productID=100038)

#### TRADEMARK CITATION

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brands and product names are registered trademarks or trademarks of their respective companies.

#### CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Curtis A. Smith  
 Defense Contract Audit Agency  
 P.O. Box 20044  
 Fountain Valley, CA 92728-0044  
 Work Phone: 714-896-4277  
 Fax: 714-896-6915  
 Email: [casmith@mindspring.com](mailto:casmith@mindspring.com)



```

/* Begin ODS output */
libname data 'c:\wuss 2002';
filename odsout 'c:\wuss 2002';
data baseball;
  set data.baseball;
  length divisiondrill $ 40;
  if league='American' and division='West' then
    divisiondrill='HREF="divisionALwest_body.html"';
  else if league='American' and division='East' then
    divisiondrill='HREF="divisionALEast_body.html"';
  else if league='National' and division='West' then
    divisiondrill='HREF="divisionNLwest_body.html"';
  else if league='National' and division='East' then
    divisiondrill='HREF="divisionNLeast_body.html"';
run;
pattern1 color=blue;
pattern2 color=green;
pattern3 color=cyan;
pattern1 color=yellow;
pattern2 color=violet;
pattern3 color=orange;
legend1 label=none shape=bar(4,4) position=(bottom left)
  offset=(-3);
ods listing close;
goptions reset=global gunit=pct xpixels=650 ypixels=600
  htitle=5 htext=4 ftitle=zapfb ftext=swiss
  colors=(yellow green red purple orange blue)
  transparency device=java noborder;
ods html body='baseball_body.html'
  (title="ODS Drill-Down Chart")
  frame='baseball_frame.html'
  contents='baseball_contents.html' path=odsout
  gtitle gfootnote style=statdoc gpath='c:\wuss 2002'
  parameters=("disabledrilldown"="true");
axis1 label=none value=none;
axis2 label=(angle=90 'Sum of Home Runs') minor=(n=1)
  offset=(0,0);
axis3 label=none split=' ';
title1 'Major League Baseball Home Run Statistics';
title2 h=2 '1986 Season';
proc gchart data=baseball;
  vbar3d division / discrete
    sumvar=no_home group=league subgroup=division
    legend=legend1 space=0 width=5 gspace=5
    maxis=axis1 raxis=axis2 gaxis=axis3 outline=black
    cframe=pink html=divisiondrill name='MLB'
    des='Major League Baseball Home Run Statistics';
run;
ods html body='divisionALwest_body.html' path=odsout;
title1 'American West';
format team $team.;
where league='American' and division='West';
vbar3d team / sumvar=no_home
  type=sum group=division subgroup=team autoref
  space=3 shape=PRISM patterned=subgroup inside=SUM
  coutline=black raxis=axis2 cframe=pink name='ALWest'
  des='American League West 1986 Home Runs';
run;
quit;
ods html close;
ods listing;

```